
FINISH 
/CLEAR 
C***************************************** 
C***   TRAVATURA RETICOLARE PIANA 
C*** 
L=20000     ! luce capannone 
H1=1500     ! altezza minima capriata 
H2=2000     ! altezza massima capriata 
NM=4        ! numero di maglie nella semi-capriata 
A1=1000      ! sezione aste briglie 
A2=500      ! sezione aste parete 
PC=10000    ! carico copertura (N/m) 
C*** NODI 
/PREP7 
N,1, 
N,NM*2+1,L 
N,NM*2+2,0,H1 
N,NM*3+2,L/2,H2 
N,NM*4+2,L,H1 
FILL,1,NM*2+1 
FILL,NM*2+2,NM*3+2 
FILL,NM*3+2,NM*4+2 
C*** 
C*** ELEMENTI 
C*** 
ET,1,1 
R,1,A1 
R,2,A2 
MP,EX,1,210000 
C***   briglia superiore 
E,1,2 
EGEN,NM*2,1,1 
C***   briglia inferiore 
E,NM*2+2,NM*2+3 
EGEN,NM*2,1,-1 
C*** aste di parete verticali 
REAL,2 
E,1,NM*2+2 
EGEN,NM*2+1,1,-1 
C***   aste di parete diagonali 
E,NM*2+2,2 
EGEN,NM,1,-1 
E,NM+1,NM*3+3 
EGEN,NM,1,-1 
C*** 
C*** VINCOLI 
C*** 
D,1,ALL,0 
D,NM*2+1,UY,0 
C*** 
C*** CARICHI 
C*** 
F,NM*2+2,FY,-PC*L/1000/(NM*2)/2    !carico sui nodi di estremità 
F,NM*2+2,FY,-PC*L/1000/(NM*2)/2 
NSEL,,NODE,,NM*2+3,NM*4+1 
F,ALL,FY,-PC*L/1000/(NM*2)      ! carico sui nodi intermedi 
NSEL,ALL 
FINISH 
C*** 
C*** SOLUZIONE 
C*** 
/SOLU 



SOLVE 
FINISH 
C*** 
C*** POST-PROCESSING 
C*** 
/POST1 
SET 
C*** opzioni di rappresentazione 
/PBC,ALL,,1 
/PBC,NFOR,,0 
/PBC,NMOM,,0 
/PBC,RFOR,,0 
/PBC,PATH,,0 
/PNUM,NODE,1 
C*** struttura con vincoli e carichi 
EPLOT 
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0 
C*** deformata 
/PNUM,NODE,0 
PLDISP,1 
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0 
C*** sforzo normale 
/PBC,ALL,,0 
ETABLE,N,SMISC,1         ! estrae il dato "sforzo normale" dal data base 
PLETAB,N                 ! rappresenta lo sforzo normale 
C*** alternativamente : PLESOL,SMISC,1 
*ASK,IFL,PREMERE RETURN PER CONTINUARE,0 
C*** tabella sforzo normale 
PRETAB,N 
 
C*** alternativamente : PRESOL,SMISC,1 
 


